

```
//Stochastics_3Lines.mq4
```

```
#property indicator_separate_window  
#property indicator_buffers 3  
#property indicator_color1 Aqua  
#property indicator_color2 Magenta  
#property indicator_color3 Gold  
  
#property indicator_maximum 100  
#property indicator_minimum 0  
  
#property indicator_level1 80  
#property indicator_level2 20
```

← #property命令を記述

```
//インジケータバッファの宣言
```

```
double K_Line[];  
double D_Line[];  
double SD_Line[];  
double Numerator[];  
double Denominator[];
```

← インジケータバッファを宣言

```
//変数の宣言
```

```
extern int K_Period = 14;  
extern int D_Period = 3;  
extern int SD_Period = 3;
```

← 変数を宣言

```
int init()
```

```
{  
    IndicatorBuffers(5);  
  
    //インジケータバッファのインデックス  
    SetIndexBuffer(0,K_Line);  
    SetIndexBuffer(1,D_Line);  
    SetIndexBuffer(2,SD_Line);  
    SetIndexBuffer(3,Numerator);  
    SetIndexBuffer(4,Denominator);  
  
    //インジケータのラベル  
    SetIndexLabel(0,"%K");  
    SetIndexLabel(1,"%D");  
    SetIndexLabel(2,"SD");  
    SetIndexLabel(3,NULL);  
    SetIndexLabel(4,NULL);  
    IndicatorShortName("Stochastics 3Lines("+K_Period+", "+D_Period+", "+SD_Period+"");  
  
    //インジケータのスタイル  
    SetIndexStyle(0,DRAW_LINE,STYLE_SOLID);  
    SetIndexStyle(1,DRAW_LINE,STYLE_SOLID);  
    SetIndexStyle(2,DRAW_LINE,STYLE_SOLID);  
    SetIndexStyle(3,DRAW_NONE);  
    SetIndexStyle(4,DRAW_NONE);  
  
    //インジケータの描画開始時点  
    SetIndexDrawBegin(0,K_Period * 2);  
    SetIndexDrawBegin(1,K_Period * 2);  
    SetIndexDrawBegin(2,K_Period * 2);  
  
    return(0);  
}
```

← 基本設定を記述

```
int start()
```

```
{  
    int limit = Bars - IndicatorCounted();  
  
    // %K  
    for(int i = 0; i < limit; i++)  
    {  
        Numerator[i] = Close[i] - Low[iLowest(NULL,0,MODE_LOW,K_Period,i)];  
        Denominator[i] = High[iHighest(NULL,0,MODE_HIGH,K_Period,i)] - Low[iLowest(NULL,0,MODE_LOW,K_Period,i)];  
  
        if(Denominator[i] != 0)  
        {  
            K_Line[i] = (Numerator[i] / Denominator[i]) * 100;  
        }  
    }  
  
    // %D  
    for(i = 0; i < limit; i++)  
    {  
        double Sum_Numerator = 0;  
        double Sum_Denominator = 0;  
  
        for(int j = i; j < i + D_Period; j++)  
        {  
            Sum_Numerator += Numerator[j];  
            Sum_Denominator += Denominator[j];  
        }  
  
        if(Sum_Denominator != 0)  
        {  
            D_Line[i] = (Sum_Numerator / Sum_Denominator) * 100;  
        }  
    }  
  
    // SD  
    for(i = 0; i < limit; i++)  
    {  
        Sum_Numerator = 0;  
  
        for(j = i; j < i + SD_Period; j++)  
        {  
            Sum_Numerator += D_Line[j];  
        }  
  
        SD_Line[i] = Sum_Numerator / SD_Period;  
    }  
  
    return(0);  
}
```

← 具体的な処理内容を記述